

## 第5章 モナドと命令型言語の意味

この章では、命令型言語のさまざまな副作用に対して、モナドを利用してその意味を Haskell で与えることにする。Haskell で意味を与えるということは、等式による推論が可能になるということである。

前章で紹介した IO は効率のために Haskell の処理系で特別扱いされる組込みのモナドであるが、他の言語の副作用を模倣するためにユーザーが独自のモナドを定義することも可能である。モナドを用いる利点は、“計算”の意味が変わっても、モナドの標準的な関数 *return* と  $(\gg=)$  のみを用いている部分は、変更する必要がないところである。

### 5.1 ミニ命令型言語 Util

この章で紹介する副作用をすべて備え持つ命令型言語は現実には存在しないので、説明のための独自のミニ命令型プログラミング言語を導入する。簡単な言語からはじめて様々な副作用を持つ言語を定義していく。状況に応じて C, Java などに加えて、このようなミニ命令型プログラミング言語のプログラムの意味を Haskell で説明する。

名前がないと不便なので、これらのミニ命令型言語を Util (**U**til: **T**iny **I**mperative **L**anguage) (いわゆる再帰的頭字語 (recursive acronym) である。PHP, GNU などの略語の由来も参照すること。)と呼び、必要により UtilErr, UtilST, UtilCont, ... などのようにバージョンを表す接尾語をつけることにする。いろいろな副作用を導入していくにつれ、その“計算”を表すモナドの定義が変わることになる。

Util については、この章では、必要な事柄の説明だけにとどめて、補足の第 U 章で実装 (Util から Haskell へのコンパイラ) の詳細を紹介する。ただし、実装の詳細は把握しなくても、ソースプログラムとターゲットプログラムを比べれば、副作用を持つプログラミング言語がどのように変換されるか感覚的に掴めるはずである。

#### 5.1.1 構文規則

Util の具体的な構文としては次のような BNF で定義されていると仮定する。  
(演算子の優先順位なども適切に宣言されているとする。)

```
Expr → Const | Var | ( Expr )
      | if Expr then Expr else Expr | while Expr do Expr
      | begin Exprs end
      | let Decls in Expr | val Binds in Expr
      | \ Vars -> Expr | Expr Expr
      | Expr + Expr | Expr * Expr | ... (他の中置演算子) ...
Exprs → Expr | Expr ; Exprs
Decl  → Vars = Expr
```

```

Decls → Decl | Decl ; Decls
Bind  → Var <- Expr
Binds → Bind | Bind ; Binds
Vars  → Var | Var Vars

```

ここに示されていないが、定数 *Const* と変数 *Var* の字句の定義は Haskell と同じとする。

Haskell と異なり **while** ~ **do** 式や **begin** ~ **end** 式があるところが命令型言語らしいところである。

### 5.1.2 抽象構文と具象構文

ところで、上の Util の構文規則は ambiguous (ambiguous) である。つまり  $1 + 2 * 3$  は足し算と掛け算のどちらを先に実行するか、決定できない。通常は曖昧さを避けるために、

```

Expr  → Expr + Term | Term
Term  → Term * Factor | Factor
Factor → Const | ( Expr )

```

のように曖昧さを避けるために構文規則を工夫する。この後者のように実際に構文解析に用いるための構文を concrete syntax (concrete syntax) という。

それに対して、いったん構文解析が終了してしまえば、曖昧さを避けるための補助的な仕掛けは必要なくなり、本質的な構造のみを扱えばよい。そのため、前者のような構文規則で十分である。このように構成要素の本質的な関係を記述した構文のことを abstract syntax (abstract syntax) という。

この章で“構文”と呼んでいるのは、この抽象構文のことである。

### 5.1.3 Util から Haskell への翻訳

命令型言語である Util から Haskell への翻訳を考えることができる。補足第 U 章では、さらに詳しく Util から Haskell へのコンパイラの実装を紹介するが、ここでは対応表の形で説明する。

次の表のソース中で *Italic* フォントで示されている  $m, n$  などは任意の Util の式で、ターゲット中で  $\llbracket m \rrbracket, \llbracket n \rrbracket$  のように  $\llbracket \rrbracket$  が付いている式は、その変換後の Haskell の式を表す。

| ソース (Util)                                                                                                          | ターゲット (Haskell)                                                                                                                                                                  |
|---------------------------------------------------------------------------------------------------------------------|----------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| <i>c</i> (ただし <i>c</i> は定数)                                                                                         | <code>return c</code>                                                                                                                                                            |
| <i>x</i> (ただし <i>x</i> は変数)                                                                                         | <code>return x</code>                                                                                                                                                            |
| <b>val</b> <i>x</i> <- <i>m</i> <b>in</b> <i>n</i>                                                                  | $\llbracket m \rrbracket >>= \backslash x \rightarrow \llbracket n \rrbracket$                                                                                                   |
| <b>let</b> <i>f</i> = $\backslash x \rightarrow m$<br><i>g</i> = $\backslash y \rightarrow n$<br><b>in</b> <i>k</i> | <b>let</b> <i>f</i> = $\backslash x \rightarrow \llbracket m \rrbracket$<br><i>g</i> = $\backslash y \rightarrow \llbracket n \rrbracket$<br><b>in</b> $\llbracket k \rrbracket$ |
| <i>f a</i>                                                                                                          | $\llbracket f \rrbracket >>= \backslash \_g \rightarrow$<br>$\llbracket a \rrbracket >>= \backslash \_x \rightarrow$                                                             |

|                                                      |                                                                                                                                       |
|------------------------------------------------------|---------------------------------------------------------------------------------------------------------------------------------------|
|                                                      | <code>_g _x</code>                                                                                                                    |
| <code>\ x -&gt; m</code>                             | <code>return (\ x -&gt; [m])</code>                                                                                                   |
| <code>if c then t else e</code>                      | <code>[c] &gt;&gt;= \ _b -&gt;<br/>if _b then [t] else [e]</code>                                                                     |
| <code>while c do t</code>                            | <code>let _while = [c] &gt;&gt;= \ _b -&gt;<br/>if _b then [t] &gt;&gt;= \ _ -&gt;<br/>_while<br/>else return ()<br/>in _while</code> |
| <code>begin<br/>  s;<br/>  t;<br/>  u<br/>end</code> | <code>[s] &gt;&gt;= \ _ -&gt;<br/>[t] &gt;&gt;= \ _ -&gt;<br/>[u]</code>                                                              |

この翻訳の要点は、変数や定数の出現など“副作用”が発生しないところには `return` がついた形に翻訳されること、関数適用や `begin ~ end` による逐次実行は `(>>=)` を使った式に翻訳されること、などである。

また、Util プログラム中の `+`, `-`, `*` などの（副作用を持たない）プリミティブな二項演算子は、他の関数が `return (\ x -> ...)` という形に変換されること、戻り値はアクションを持たないことから、同名の Haskell 内のオペレーターを用いて、それぞれ

```
return (\ x -> return (\ y -> return (x + y)))
return (\ x -> return (\ y -> return (x - y)))
return (\ x -> return (\ y -> return (x * y)))
```

という Haskell の式に置換するようにしておくとの部分の変換と整合する。

| ソース (Util)                    | ターゲット (Haskell)                                                                 |
|-------------------------------|---------------------------------------------------------------------------------|
| ⊗<br>(ただし ⊗ は<br>副作用のない二項演算子) | <code>return (\ x -&gt;<br/>  return (\ y -&gt;<br/>    return (x ⊗ y)))</code> |

この対応表を用いて、例えば次の Util プログラムを変換 (ただし、この `fact` 関数は副作用を含んでいないので、この変換自体にはあまり意味はない。) すると

```
1 fact n = if n == 0 then 1 else n * fact (n - 1)
```

次のような Haskell のプログラムが得られる。

```
1 fact = \ n ->
2   ((return (\ x -> return (\ y -> return (x == y))) >>=
3     \ _f -> return n >>= \ _x -> _f _x)
4     >>= \ _f -> return 0 >>= \ _x -> _f _x)
5     >>=
6     \ _b ->
7       if _b then return 1 else
8         (return (\ x ->
9           return (\ y -> return (x * y))) >>=
10          \ _f -> return n >>= \ _x -> _f _x)
11          >>=
12          \ _f ->
```

```

13         (return fact >>=
14           \ _f ->
15             ((return (\ x ->
16               return (\ y -> return (x - y)))
17               >>= \ _f -> return n
18                 >>= \ _x -> _f _x)
19                 >>= \ _f -> return 1
20                   >>= \ _x -> _f _x)
21                   >>= \ _x -> _f _x)
22                   >>= \ _x -> _f _x)

```

これは多くの冗長な部分を含んでいるので、前述の monad law などを利用して単純化すると、多くの `return` や `(>>=)` が消えて、次のような Haskell の式が得られる。

```

1 fact = \ n -> if n == 0 then return 1 else
2           fact (n - 1) >>= \ _x ->
3           return (n * _x)

```

## 5.2 トリビアルなバージョン — Utlil

最初のバージョン Utlil では、モナドはトリビアルな計算（何もしない計算）としておく。つまり、Utlil は副作用を持たない言語である。

ファイル `Id.hs`

```

1 newtype I a = I a
2
3 instance Functor I where
4   fmap f m = m >>= \ x -> return (f x)
5
6 instance Applicative I where
7   pure a = I a
8   g <*> m = g >>= \ f -> m >>= \ x -> return (f x)
9
10 instance Monad I where
11   (I m) >>= k = k m

```

ある型構成子を `Monad` のインスタンスと宣言するとき、同時に `Monad` のスーパークラスである `Functor` と `Applicative` に対してもインスタンス宣言をする必要がある。`I` に対するインスタンス宣言だけは `fmap` と `<*>` の定義も示しておく。

ここで、newtype は、Haskell の新しい型の宣言の形式の一つである。data 宣言と似ているが、フィールドが一つの構成子を一つしか持つことができない。また data 宣言の構成子と異なり、newtype 宣言で導入される構成子は型変換の意味しか持たず、実行時の計算を伴わない。また、型の別名を宣言する type 宣言との違いは、型の変換が構成子によって明示的になる点である。（型クラスのインスタンスには、type 宣言で導入された型の別名は指定できない。例えばリストに通常の辞書式順序と異なる順序（Ord のインスタンス宣言）を与えたいときは newtype を使って別名を用意し、別名に対してインスタンス宣言する必要がある）

上記の `newtype` 宣言では型構成子と構成子に同じ `I` という名前を使っているが、文脈でどちらか判断することができるので問題ない。

このとき、

```
1 fact n = if n == 0 then 1 else n * fact (n - 1)
```

という `UtilI` プログラムを Haskell に翻訳して実行する `unI (fact 9)` と、同じプログラムを Haskell として実行したときと全く同じ 362880 という値になる。これは何の意味もないように思えるが、副作用がなければ、`Util` と Haskell は意味が変わらないという確認にはなる。

ただし、`unI` は次のように定義された型変換関数である。

ファイル `Id.hs`

```
1 unI :: I a -> a
2 unI (I a) = a
```

## 5.3 UtilST — 状態の導入

`Util` に更新（破壊的代入）可能な状態の概念を導入する。ここで紹介する例では、2 つだけ代入可能な参照 (reference) `xP` と `yP` を導入することにする。2 という数は別に本質的なものではなく、いくつにすることも可能である。参照は `:=` 演算子で値を代入し、`get` で値を取り出すことができる。

例えば、

```
1 begin xP := 1; xP := get xP + 3; get xP end
```

という `UtilST` プログラムを評価すると、`__` という結果が得られる。

**参考:**なお、`Util` に代入可能な変数をそれ以外の変数と区別するような宣言（例えば Kotlin の `var` と `val`）を導入したり、変数名のカテゴリーを区別（例えば、特定の文字セットから始まる変数名を代入可能な変数に割り当てる）したりすれば、`get` の適用を省略して、

```
1 begin μ := 1; μ := μ + 3; μ end
```

のように書くことができるように言語仕様を変更することも可能である。こうすれば、C 言語により近い記法でプログラムを記述することも可能である。以下では、`μ`、`ν` のようにギリシア文字を使った変数は、代入可能な参照を指す変数として宣言されていると仮定する。しかし `get` を使う書き方にも、副作用が存在している箇所が明確になると利点がある。

状態を導入するために、やはり “計算の型” を定義する必要がある。まず次の `ST` という型構成子を定義する。

ファイル ST.hs

```
1 newtype ST s a = ST (s -> (a,s))
2
3 unST :: ST s a -> s -> (a,s)
4 unST (ST s) = s
5
6 instance Applicative (ST s) where
7   pure a = ST (          )
8   (<*>) = ap
9
10 instance Monad (ST s) where
11   (ST m) >>= k = ST (\ s0 -> let { (a,s1) = m s0 }
12                               in unST (k a) s1)
13   -- ST, unST がなければ、次のようになる
14   -- m >>= k = \ s0 -> let (a,s1) = m s0 in k a s1
```

この型構成子は `ST s` という形で `s` というパラメーター付きのモナドになる。このモナドは State Transformer モナドと呼ばれる。 `return a` は状態 (`s`) の変更を行わず、`a` をそのまま返す計算である。このモナドの `m >>= k` は、`m` で変更された状態 (`s1`) をそのまま、`k` に受渡す計算である。

**問 5.3.1** `ST` に対して、次の monad law が成り立つことを確認せよ。

```
(return a) >>= k = k a
m >>= (\ a -> return a) = m
(m1 >>= k1) >> k2 = m1 >>= (\ a -> (k1 a >>= k2))
```

参照は次のように定義する。

ファイル MyState.hs

```
1 type Pos s a = s -> (a, a -> s)
2
3 xP :: Pos (x,y) x
4 xP = \ (x,y) -> (x, \ x1 -> (x1,y))
5
6 yP :: Pos (x,y) y
7 yP = \ (x,y) -> (y, \ y1 -> (x,y1))
```

`Pos s a` は `s` という全体の状態の型の中で、`a` という型を指している参照の型である。ここで `xP` と `yP` は、それぞれペアの第 1、第 2 成分にアクセスするための参照である。参照に対する読み出し・書き込みのメソッドは、後で別のモナドでも使用するので、型クラス `MyState` のメソッドとして定義しておくことにする。

ファイル MyState.hs

```
1 class MyState m where
2   get :: Pos s a -> m s a
3   set :: Pos s a -> a -> m s a
```

そして `ST` をこの `MyState` クラスのインスタンスとして宣言する。

ファイル ST.hs

```

1 instance MyState ST where
2   get p  = ST (\ s -> (fst (p s), s))
3   set p v = ST (\ s -> (v, snd (p s) v))
4
5 -- 例えば get xP ≡ ST (\ (x,y) -> (x, (x,y)))
6 --          set xP x1 ≡ ST (\ (x,y) -> ((x1,y)))

```

ここで `set` は参照先の状態を書き換える、また `get` は参照先の値を複製する関数である。

また、最終的に `ST` から値を取り出すために

```

1 evalST :: ST s a -> s -> a
2 evalST st s = fst (unST st s)

```

という関数を定義しておく。第2引数の `s` は初期状態である。

**Q 5.3.2** 次のように `ST` を使った関数 `foo, bar` を定義する。

```

1 import ST
2 import MyState
3
4 foo b y = do set xP 1
5              set yP y
6              let repeat =
7                  do y1 <- get yP
8                     if y1 > 0 then do
9                         x1 <- get xP
10                        set xP (x1 * b)
11                        set yP (y1 - 1)
12                        repeat
13                     else return ()
14              in repeat
15              get xP
16
17 bar n = do set xP 1
18            set yP 1
19            let repeat =
20                do x1 <- get xP
21                   if x1 < n then do
22                       y1 <- get yP
23                       y2 <- set yP (y1 + 1)
24                       set xP (x1 * y2)
25                       repeat
26                   else return ()
27            in repeat
28            get yP

```

これらの関数に対して、

1. `evalST (foo 3 4) (0,0)` \_\_\_\_\_

2. `evalST (bar 50) (0,0)` \_\_\_\_\_

の値を、まず実行する前に予想し、次に実際に実行して確認せよ。

`UtilST` の `:=` 演算子と `get` 関数は、Haskell の `set`, `get` にそれぞれコンパイルされるようにしておく。

| ソース (Util)      | ターゲット (Haskell)                                                                                                                           |
|-----------------|-------------------------------------------------------------------------------------------------------------------------------------------|
| $p := m$        | $\llbracket p \rrbracket \gg= \lambda \_p \rightarrow$<br>$\llbracket m \rrbracket \gg= \lambda \_x \rightarrow$<br>$\text{set } \_p \_x$ |
| $\text{get } p$ | $\llbracket p \rrbracket \gg= \lambda \_p \rightarrow$<br>$\text{get } \_p$                                                               |

左の UtilST プログラム（右は対応する C プログラム）：

|                                                                                                                                                                                |                                                                                                                                    |
|--------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|------------------------------------------------------------------------------------------------------------------------------------|
| <pre> 1 fact n = begin 2   xP := 1; yP := n; 3   while get yP &gt; 0 do 4     begin 5       xP := get xP * get yP; 6       yP := get yP - 1 7     end; 8   get xP 9 end </pre> | <pre> 1 int fact(int y) { 2   int x = 1; 3   while (y &gt; 0) { 4     x = x * y; 5     y = y - 1; 6   } 7   return x; 8 } 9 </pre> |
|--------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|------------------------------------------------------------------------------------------------------------------------------------|

**参考:** 次は `get` を省略する書き方にして、さらに C に近付けた UtilST プログラムである。

|                                                                                                                                       |
|---------------------------------------------------------------------------------------------------------------------------------------|
| <pre> 1 fact n = begin 2   μ := 1; ν := n; 3   while ν &gt; 0 do begin 4     μ := μ * ν; 5     ν := ν - 1 6   end; 7   μ 8 end </pre> |
|---------------------------------------------------------------------------------------------------------------------------------------|

を翻訳すると次のような Haskell の関数になる。

|                                                                                                                                                                                                                                                                                                                                                                                                                                                                  |
|------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| <pre> 1 fact n = set xP 1 &gt;&gt;= \ _ -&gt; 2   set yP n &gt;&gt;= \ _ -&gt; 3     (let _while = get yP &gt;&gt;= \ y -&gt; 4       if y &gt; 0 then 5         get xP &gt;&gt;= \ x -&gt; 6         get yP &gt;&gt;= \ y -&gt; 7         set xP (x * y) &gt;&gt;= \ _ -&gt; 8         get yP &gt;&gt;= \ y -&gt; 9         set yP (y - 1) &gt;&gt;= \ _ -&gt; 10        _while 11      else return () 12    in _while) &gt;&gt;= \ _ -&gt; 13    get xP </pre> |
|------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|

この翻訳した `fact` に対して `fact 9` を実行する `(evalST (fact 9) (0,0))` と、その結果は 362880 になる。

階乗の場合、普通に関数的な定義のほうが簡潔だが、変数の数が多い場合などは、このような命令的な書き方が簡潔になる場合も考えられる。

この ST の定義では、エラー処理を考慮していない。エラー処理を行なうためには、この ST と（後述する）Maybe の定義を合成する必要がある。参考まで

に、次のようなモナドになる。

ファイル `EST.hs`

```
1 newtype EST s a = EST (s -> Maybe (a,s))
2
3 unEST :: EST s a -> s -> Maybe (a,s)
4 unEST (EST m) = m
5
6 instance Applicative (EST s) where
7   pure a = EST (\ s -> Just (a, s))
8   (<*>) = ap
9
10 instance Monad (EST s) where
11   (EST m) >>= k = EST (\ s0 ->
12     case m s0 of
13       Just (a,s1) -> unEST (k a) s1
14       Nothing      -> Nothing)
```

**問 5.3.3** 次の C の関数とほぼ同等な `UtilST` の関数、または、`ST` モナドを用いた Haskell の関数を定義せよ。

```
1. 1 int foo(int n) {
   2   int i = 1, j = 1;
   3   while (i < n) {
   4     i = i + j;
   5     j = i - j;
   6   }
   7   return i;
   8 }

2. 1 int bar(int n) {
   2   int i = 0;
   3   while (n > 1) {
   4     i = i + 1;
   5     n = n / 2;
   6   }
   7   return i;
   8 }
```

※ 整数の割り算は Haskell では ``div`` 演算子になる。

## 5.4 (参考) `UtilIO` — 入出力の導入

入出力は、入出力ストリームを状態の一種と考えれば、前節の `UtilST` と同じ方法で取り扱うことができる。

計算のモナドの定義は前節と基本的に同じだが、状態に入力と出力のストリームを表す `String` 型の部分を追加しておく。

ファイル `MyStream.hs`

```
1 type WithIO s = (s,String,String)
```

第2要素が入力ストリーム、第3要素が出力ストリームを表す部分である。

ファイル MyIO.hs

```
1 newtype MyIO s a = MyIO (WithIO s -> (a, WithIO s))
2
3 unMyIO :: MyIO s a -> WithIO s -> (a, WithIO s)
4 unMyIO (MyIO m) = m
```

参照のプリミティブの定義は次のようになる。

ファイル MyIO.hs

```
1 instance MyState MyIO
2   get p    = MyIO (\ (s,i,o) -> (fst (p s), (s,i,o)))
3   set p v = MyIO (\ (s,i,o) -> ((), (snd (p s) v, i, o)))
```

入出力に関するプリミティブも後で別のモナドで使用するので、型クラス `MyStream` のメソッドとして定義しておく。

ファイル MyStream.hs

```
1 class MyStream m where
2   readChar  :: m Char
3   eof       :: m Bool
4   writeStr  :: String -> m ()
```

ファイル MyIO.hs

```
1 instance MyStream (MyIO s) where
2   readChar  = MyIO (\ (s,c:cs,o) -> (c, (s,cs,o)))
3   eof       = MyIO (\ (s,i,o) -> (null i, (s,i,o)))
4   writeStr v = MyIO (\ (s,i,o) -> ((), (s,i,o ++ v)))
```

ここで `readChar` は入力ストリームから1文字を取出す。また `writeStr str` は出力ストリーム `o` に `str` を追加する (ただし「++」の計算量は左オペランドの長さに比例するので、この定義のように文字列の後ろに新しい文字列を追加 (++) していくと、出力文字列が長くなるにしたがって効率が悪くなる。これを避けて効率の良い定義を与えることも可能であるが、ここでは簡単のために「++」を使った定義を採用する。)。そして `write` という関数を次のように定義しておく。

ファイル MyStream.hs

```
1 write :: (Show v, MyStream m) => v -> m ()
2 write v = writeStr (show v)
```

すると、次の `UtilIO` プログラム (ただし、「//」は整数の除算を表す演算子とする。)

```
1 foo n = begin
2   xP := n;
3   while get xP > 0 do begin
4     write (get xP % 10);
5     xP := get xP // 10
6   end
7 end
```

を Haskell に翻訳した結果は、

```
1 foo n = set xP n >>= \ _ ->
2     let _while
3         = get xP >>= \ _x ->
4             if _x > 0 then
5                 get xP >>= \ _x ->
6                     write (_x `mod` 10) >>= \ _ ->
7                         get xP >>= \ _x ->
8                             set xP (_x `div` 10) >>= \ _ ->
9                                 _while
10                            else return ()
11     in _while
```

となり、補助関数 evalMyIO を以下のように定義したとき、

```
1 evalMyIO e s =
2     let (_, (_, _, o)) = unMyIO e (s, "", "") in o
```

関数 foo を `evalMyIO (foo 12345) (0,0)` のように実行すると、出力は "54321" になる。

**問 5.4.1** 次の C の関数とほぼ同等な UtilIO の関数、または、MyIO モナドを用いた Haskell の関数を定義せよ。

1. 

```
1 int baz(int n) {
2     int i, j;
3     for (i = 0; i < n; i++) {
4         for (j = 0; j <= i; j++) {
5             printf("*");
6         }
7         printf("\n");
8     }
9     return i;
10 }
```
2. 

```
1 int qux(int n) {
2     int i, j;
3     for (i = 0; i < n; i++) {
4         printf("*");
5         if (i % 3 == 0) {
6             printf("!");
7         }
8         printf("\n");
9     }
10    return i;
11 }
```

## 5.5 UtilErr — エラー処理の導入

次に Util にエラー処理を導入する。Err という接尾語をつけて UtilErr と呼ぶことにする。UtilErr は、生真面目に (?) エラー処理を行ない、部分式にエラーがあれば式全体もエラーになるようにする。このように UtilErr は (Haskell のような) 遅延評価ではなくて、関数の引数を必ず先に評価する \_\_\_\_\_ (eager evaluation)、つまり普通の言語の評価順を模倣する。

エラーと正常な振舞いを区別するために、次のような標準ライブラリーに用意されているデータ型 `Maybe` を使用する。

```
1 -- Preludeに定義済み
2 data Maybe a =      |      deriving (Show,Eq)
```

正常な振舞いは `Just` という構成子で表す。エラーの場合は `Nothing` という構成子を用いる。この型に対して次のようなインスタンス宣言がされている。

```
1 -- Preludeに宣言済み
2 instance Applicative Maybe where
3   pure a = Just a
4   (<*>) = ap
5
6 instance Monad Maybe where
7   (Just a) >>= k =
8   Nothing  >>= k =
```

このモナドの `m >>= k` は、まず `m` を計算し、その計算が正常終了すれば、その値を `k` という関数に渡す。しかし、いったん `m` でエラーが起こると、`k` は評価されず、\_\_\_\_\_ していくことを表している。

**問 5.5.1** `Maybe` に対して、次の `monad law` が成り立つことを確認せよ。

```
(return a) >>= k = k a
m >>= (\ a -> return a) = m
(m1 >>= k1) >> k2 = m1 >>= (\ a -> (k1 a >>= k2))
```

さらに、`MonadPlus` というクラスに属するいくつかのメソッドを用意する。ここで `mzero` は \_\_\_\_\_ 状況をつくり出すときに用いる。

`Maybe` に対する `mplus` は第 1 引数を評価し、\_\_\_\_\_ 第 2 引数を評価する関数である。

```
1 -- モジュール Control.Monad に定義済み
2 class Monad m => MonadPlus m where
3   mzero :: m a
4   mplus :: m a -> m a -> m a
5
6 -- モジュール Control.Monad に宣言済み
7 instance MonadPlus Maybe where
8   mzero = Nothing
9   Just a `mplus` _ = Just a
10  Nothing `mplus` m = m
```

**Q 5.5.2** 次のように `Maybe` を使った関数 `foo, bar` を定義する。

```
1 import Control.Monad
2
3 mydiv :: Double -> Double -> Maybe Double
4 x `mydiv` y = if y == 0 then mzero else return (x / y)
5
6 foo :: Double -> Maybe Double
7 foo n = do x <- 6 `mydiv` n
8         return (x * 2)
```

```

9
10 bar :: Double -> Double -> Maybe Double
11 bar m n = do x <- (4 `mydiv` m) `mplus` (return 3)
12             x `mydiv` n
13

```

これらの関数に対して、(1) `foo 2` (2) `foo 0` (3) `bar 0 4` (4) `bar 1 0` の値を、まず実行する前に予想し、次に実際に実行して確認せよ。

いわゆるプリミティブ関数もエラー処理を利用するように書き換えることができる。例えば、`UtilErr` の割り算 (`/`) は、0 で割るとエラーなので、`mzero` を明示的に使って、次のような Haskell の式に置換されるようにする。

```

\ x -> return (\ y -> if y == 0 then mzero
                      else return (x / y))

```

これで 0 で割ろうとした場合にはエラーが報告される。

例えば

```

1 (\ x -> 999) (1 / 0)

```

のような式は、Haskell の式としても `Util` の式としても解釈できるが、Haskell だと解釈すると、`1 / 0` の部分式は \_\_\_\_\_ に全体の結果が 999 となるが、`UtilErr` だと解釈すると、次のような Haskell プログラムに翻訳され、

```

1 (if 0 == 0 then mzero
2   else return (1 / 0)) >=> \ _x ->
3 return 999

```

実行すると（エラーが起こったことを表す） \_\_\_\_\_ という結果になる。つまり、先行評価をシミュレートしていることになる。

## 5.6 例外処理の導入

例外のモナド `Maybe` を利用して、Java の `try ~ catch` のように例外を捕捉する構文を導入することも可能である。

`Util` の BNF には以下の構文を追加する。

*Expr* → ... | **try** *Expr* **catch** *Expr*

“**try** *m* **catch** *n*” は *m* を評価し、エラーがなかった場合は、その戻り値を **try** 式の戻り値とする。しかし *m* の評価中にエラーが生じた場合は、*n* を評価する。`Util` の “**try** *m* **catch** *n*” は “`[[m]] `mplus` [[n]]`” という式として構文解析されるようにしておく。

また、`fail` という `Util` の関数は、Haskell の `mzero` を返す関数にコンパイルされるようにしておく。この関数は、Java の `throw` 文に対応する。

|            |                 |
|------------|-----------------|
| ソース (Util) | ターゲット (Haskell) |
|------------|-----------------|

|                            |                                  |
|----------------------------|----------------------------------|
| <code>try m catch n</code> | <code>[[m]] `mplus` [[n]]</code> |
| <code>fail ()</code>       | <code>mzero</code>               |

例えば

```
1 bar n = try 1 / n catch 99999
```

という UtilErr プログラムをコンパイルすれば、出力される Haskell プログラムは、

```
1 bar n = (if n == 0 then mzero else return (1 / n))
2         `mplus` (return 99999)
```

であり、`bar 0` の結果は `Just 99999.0` である。

**問 5.6.1** 次の Java のメソッドとほぼ同等な Haskell の関数をモナドを用いて作成せよ。

```
1 public static void hoge(int n) {
2     int i;
3     for (i = -n; i <= n; i++) {
4         try {
5             System.out.printf("%d ", 360 / i);
6         } catch (Exception e) {
7             System.out.printf("/ by 0 ");
8         }
9     }
10    System.out.println();
11 }
```

## 5.7 UtilNonDet — 非決定性の導入

ここからは対象言語 Util に非決定性を導入する。非決定性 (nondeterminism) とはプログラムの動作に                      が存在することを言う。ある選択肢を選んだ結果、計算が失敗する場合がある、その場合は前の選択肢に戻って計算をやり直す (バックトラック)。非決定性は探索型のパズル・ゲームや構文解析プログラムなどで利用できる。バックトラックをプリミティブな機能として提供する言語としては、論理型言語の            が有名である。

非決定性の“計算”のモナドは通常のリストを使用する。

```
1 -- Prelude で宣言済み
2 instance Applicative [] where
3     pure a = [a]
4     (<*>) = ap
5
6 instance Monad [] where
7     [] >>= k = []
8     (x:xs) >>= k = k x ++ (xs >>= k)
9
10 -- モジュール Control.Monad で宣言済み
11 instance MonadPlus [] where
12     mzero = []
13     mplus = (++)
```

このモナドは複数の選択肢を単にリストとして表現している。要するに、失敗は空リスト `[]` と表し、“ふつつ”の成功は、シングルトン（要素数1のリスト）で表している。やり直しを「`++`」で表している。つまり、`Maybe` のときと違って、将来ダメだったときの選択を残している。

**Q 5.7.1** 次のように `Maybe` とリストを使った関数 `foo, bar` を定義する。

（`foo, bar` は定義は同一で、型だけが異なる。）

```
1 import Control.Monad
2
3 -- mydiv は型を明示しない
4 x `mydiv` y = if y == 0 then mzero else return (x / y)
5
6 foo :: Double -> Maybe Double
7 foo m = do x <- (12 `mydiv` m) `mplus` (12 `mydiv` (m
8           + 1))
9           12 `mydiv` (4 - x)
10
11 bar :: Double -> [Double]
12 bar m = do x <- (12 `mydiv` m) `mplus` (12 `mydiv` (m
13           + 1))
14           12 `mydiv` (4 - x)
```

これらの関数に対して、① `foo 4` ② `bar 4` ③ `foo 3` ④ `bar 3` の値を、まず実行する前に予想し、次に実際に実行して確認せよ。

**問 5.7.2** `[]` に対して、次の monad law が成り立つことを確認せよ。

```
(return a) >>= k = k a
m >>= (\ a -> return a) = m
(m1 >>= k1) >> k2 = m1 >>= (\ a -> (k1 a >>= k2))
```

実はリストに対する `return` と `(>>=)` は、リストの内包表記を説明するときに使った `unit :: a -> [a]` と `bind :: [a] -> (a -> [b]) -> [b]` と全く同一の関数である。

計算の失敗は空リストで表される。

`UtilNonDet` は、構文は `UtilErr` と同じである。つまり、以下の構文を持つ。

*Expr* → ... | try *Expr* catch *Expr*

`UtilNonDet` の `try m catch h` は、`m` を評価し、`h` は“バックトラック”が起きたときに評価される。

`UtilNonDet` プログラム

```
1 test0 = (try 2 catch 3) * (try 5 catch 7)
```

をコンパイルすると、次の Haskell プログラムが得られる。

```
1 test0 = (return 2 `mplus` return 3) >>= \ x ->
2         (return 5 `mplus` return 7) >>= \ y ->
```

```
3      return (x * y)
```

この、test0 は `[ x * y | x <- [2,3], y <- [5,7]]` というリスト内表記と同じ意味になる。そして test0 は、`[10,14,15,21]` となる。

また、次の UtilNonDet プログラム

```
1 test1 n = (try 1 catch 2) / (try n catch 4)
```

を翻訳すると、次の Haskell プログラムが得られる。

```
1 test1 n = (return 1 `mplus` return 2) >>= \ x ->
2           (return n `mplus` return 4) >>= \ y ->
3           if y == 0 then mzero else return (x / y)
```

そして test1 2 は、`[0.5,0.25,1.0,0.5]`, test1 0 は `[0.25,0.5]` となる。失敗している計算については結果に現れていないことに注意する。同じプログラムを UtilErr で翻訳すると、UtilErr ではバックトラッキングが起これないので、test1 0 は全体が失敗 (Nothing) に終わる。

なお、次の head を用いてリストの頭部を取ることににより、成功した最初の計算だけを返すことも可能である。

```
1 -- Prelude に定義済み
2 head :: [a] -> a
3 head (x:_) = x
```

このとき head (test1 0) の値は 0.25 となる。この場合、Haskell が           を採用しているため、他の選択枝の計算は行なわれない。そのため選択枝が無数個あるような場合（再帰的に定義するとありうる）でも最初の選択枝の計算結果を出力することができる。

8 クイーンの問題も非決定性を用いて記述することができる。

```
1 safe1 xs n m = null xs ||
2                 val y <- head xs;
3                 ys <- tail xs in
4                 y /= n && y /= n + m && y /= n - m
5                 && safe1 ys n (m + 1);
6
7 safe xs n = safe1 xs n 1;
8
9 range i j = if i > j then fail ()
10             else try i catch range (i + 1) j;
11
12 queen n = if n == 0 then []
13            else val p <- queen (n - 1);
14                n <- range 1 8 in
15                if safe p n then (n:p) else fail ()
```

そして queen 8 は、`[4,2,7,3,6,8,5,1]` から始まる 92 個の解を返す。

**問 5.7.3** 非決定性と状態の両方の特徴を持つ計算の型として、

```

1 newtype STL s a = STL (s -> ([a],s))
2 newtype LST s a = LST (s -> [(a,s)])

```

の2つのバリエーションが考えられる。このそれぞれに対して、コンパイラーの定義を完成させ、2つの違いを説明せよ。

## 5.8 Prolog と論理変数

論理型言語の Prolog には非決定性の他に、          という特徴がある。論理変数 (logical variable) とは、最初は値が定まっておらず、単一化の制約により徐々に値が具体化していく変数である。何度も代入できる命令型言語の変数とも、一度初期化すれば二度と代入ができない関数型言語の変数とも異なる。Haskell では論理変数自体は、破壊的代入を模倣するために使った State Transformer モナドと同様の方法で模倣することができる。

例えば、Prolog では、リストを接続 (append) するプログラムは次のように記述される。

```

1 myappend([H|X], Y, [H|Z]) :- myappend(X, Y, Z).
2 myappend([], Y, Y).

```

これは、1 番目の引数と 2 番目の引数を接続した結果が 3 番目の引数になる、という関係を表している。

この myappend に対して、[1,4,3] と [5,3] の接続を求めるには、次のような問合せ（呼出し）をする。

```

1 - myappend([1,4,3], [2,5], R).
2
3 R = [1,4,3,2,5] ;
4
5 No

```

さらに Prolog のおもしろいところは myappend の逆向きの計算もできるということである。

```

1 ?- myappend(A, B, [1,2,3]).
2
3 A = []
4 B = [1,2,3] ;
5
6 A = [1]
7 B = [2,3] ;
8
9 A = [1,2]
10 B = [3] ;
11
12 A = [1,2,3]
13 B = [] ;
14
15 No

```

この実行例では接続して [1,2,3] になる 2 つのリストの、すべての可能性を求めていることになる。ユーザーが「;」を入力するたびにバックトラッキングが

起こり別解を表示する。

MicroKanren.hs (<https://github.com/rntz/ukanren>) は、 $\mu$ Kanren という論理プログラミングのための埋め込み言語 (embedded language) を実装するための Haskell のライブラリーである。 $\mu$ Kanren は miniKanren という言語のミニマルな実装である。このライブラリーでは、非決定性と論理変数にプラスアルファの機能を加えたモナドが定義されている。ここで詳細な説明には立ち入らないが、このモナドは State Transformer と非決定性を組み合わせたものである。このライブラリーを使えば、上の Prolog の myappend プログラムに相当するプログラムを Haskell で次のように記述できる。

```
1 myAppend a b ab =
2     do { h <- fresh; t <- fresh; res <- fresh;
3         ht <- cons h t; ht === a;
4         hres <- cons h res; hres === ab;
5         myAppend t b res }
6     `mplus`
7     do { n <- nil; n === a; b === ab }
```

ここで、fresh は新しい論理変数を生成する関数であり、「===」は両辺が単一化されることを示す演算子である。

ちなみに、Util の文法では次のように書くことができる。

```
1 myAppend a b ab =
2     try val h = fresh() in val t = fresh() ;
3     res = fresh() in
4     begin
5         cons h t === a; cons h res === ab;
6         myAppend t b res
7     end
8     catch begin nil() === a; b === ab end
```

この myAppend に対して次のようなプログラムを実行する。ただし、toLVList は通常のリストの型から論理変数を含むリストの型への変換、fromLVList はその逆である。ただし fromLVList c xs は xs のなかの未定の論理変数を c に置き換えた通常のリストを返す。

```
1 exampleAppend a =
2     val xs = fresh();
3     ys = fresh();
4     zs = toLVList [1,2,3] in begin
5         myAppend xs ys zs;
6         (fromLVList 0 xs, fromLVList 0 ys)
7     end
```

その結果は [([], [1,2,3]), ([1], [2,3]), ([1,2], [3]), ([1,2,3], [])] になる。

---

## 5.9 さらに詳しく知りたい人のために...

Parsec (Leijen & Meijer 2001) はモナドを利用した有名なパーサーライブラリーである。(Wadler 1992a) はモナドを用いてインタプリターを構築する方法を解説している。(Wadler 1992b) にも、モナドを用いてパーサーを構築する技法の解説がある。(Hinze 1998) は、Prolog のカット等のオペレーターの意味を整理している。(Kiselyov et al. 2005) は、miniKanren を Haskell で実装するためのモナドの解説である。

**(Leijen & Meijer 2001)** Daan Leijen and Erik Meijer, "Parsec: Direct Style Monadic Parser Combinators for the Real World"  
Technical Report UU-CS-2001-35, Dept. of Comp. Sci, Universiteit Utrecht,  
2001 年, <http://www.cs.uu.nl/people/daan/parsec.html>

**(Wadler 1992a)** Philip Wadler, "The essence of functional programming"  
19th Annual Symposium on Principles of Programming Languages (invited talk), 1992 年 1 月

**(Wadler 1992b)** Philip Wadler, "Monads for functional programming"  
Program Design Calculi, Proceedings of the Marktoberdorf Summer School, 1992 年 7–8 月

**(Hinze 1998)** Ralf Hinze, "Prological Features in a Functional Setting  
Axioms and Implementations"  
Third Fuji International Symposium on Functional and Logic Programming,  
1998 年

**(Kiselyov et al. 2005)** Oleg Kiselyov, Chung-chieh Shan, Daniel P.  
Friedman and Amr Sabry, "Backtracking, interleaving, and terminating  
monad transformers (functional pearl)"  
ACM SIGPLAN Notices (Vol. 40, No. 9, pp. 192–203), ACM, 2005 年 9 月

---

